

Е. А. СИДОРОВА, С. П. ЖЕЛЕЗНЯК

ОБРАБОТКА ОДНОМЕРНЫХ МАССИВОВ НА VBA

ОМСК 2022

Министерство транспорта Российской Федерации
Федеральное агентство железнодорожного транспорта
Омский государственный университет путей сообщения

Е. А. Сидорова, С. П. Железняк

ОБРАБОТКА ОДНОМЕРНЫХ МАССИВОВ НА VBA

Утверждено методическим советом университета
в качестве учебно-методического пособия
к выполнению самостоятельной и лабораторных работ

Омск 2022

УДК 004.42 (075.8)
ББК 32.973-018.2я73
С34

Обработка одномерных массивов на VBA: Учебно-методическое пособие к выполнению самостоятельной и лабораторных работ / Е. А. Сидорова, С. П. Железняк; Омский гос. ун-т путей сообщения. Омск, 2022. 23 с.

Учебно-методическое пособие разработано в соответствии с рабочими программами дисциплин информационного профиля с учетом требований ФГОС ВО последнего поколения.

Приведены краткие теоретические сведения по программированию задач обработки массивов данных на VBA. Рассмотрены основные этапы выполнения заданий, приведены примеры графических схем алгоритмов и листинги программ решения поставленных задач. Представлены индивидуальные практические задания, контрольные и тестовые вопросы.

Предназначено для самостоятельной и лабораторных работ студентов всех направлений подготовки (специальностей) очной и заочной форм обучения по дисциплинам, изучающим основы программирования.

Библиогр.: 5 назв. Табл. 5. Рис. 6.

Рецензенты: доктор техн. наук, профессор А. В. Бубнов;
доктор техн. наук, профессор А. А. Кузнецов.

ОГЛАВЛЕНИЕ

Введение	5
1. Общие требования к выполнению заданий	6
2. Понятие массива данных и общие принципы обработки его элементов	8
3. Определение максимального или минимального элемента массива	12
4. Задания	17
5. Контрольные вопросы.....	20
6. Примеры тестовых вопросов.....	20
Библиографический список.....	22

ВВЕДЕНИЕ

Программирование на языке *Visual Basic for Applications* (VBA) позволяет расширить возможности приложений *Microsoft Office*. Инструментальная среда VBA, реализованная в виде полнофункционального встроенного редактора, обеспечивает удобный интерфейс, наличие всех необходимых средств управления программным кодом и большое количество встроенных готовых объектов. Это позволяет эффективно применять VBA для автоматизации задач, связанных с обработкой различных структур данных.

В пособии приведены краткие теоретические сведения по алгоритмизации и программированию задач обработки одномерных массивов данных на VBA. Рассмотрены основные этапы выполнения заданий, приведены примеры графических схем алгоритмов и листинги программ решения поставленных задач. Представлены индивидуальные задания, контрольные и тестовые вопросы.

Библиографический список, приведенный в конце пособия, содержит литературу для углубленного изучения материала по рассматриваемой тематике.

1. ОБЩИЕ ТРЕБОВАНИЯ К ВЫПОЛНЕНИЮ ЗАДАНИЙ

В каждой лабораторной или самостоятельной работе необходимо выполнить следующие действия.

1. Создать рабочую книгу Excel. В свойствах файла в поле *Название* указать свои фамилию и группу, например, Иванов_40а. Сохранить рабочую книгу с поддержкой макросов с именем, указанным в [табл. 1.1](#).

2. Создать в книге Excel в редакторе VBA стандартный модуль Module1 (переименовывать его не нужно). В разделе общих объявлений (в начале) модуля ввести оператор `Option Explicit` для запрета использования необъявленных переменных. Далее в этом модуле записывать программы всех заданий текущей работы.

3. Каждое задание выполнить в следующем порядке.

3.1. Записать в тетрадь условие задачи индивидуального варианта (ИВ).

3.2. Вручную изобразить в тетради графическую схему алгоритма (ГСА) решения задачи ИВ.

3.3. В Excel-файле создать рабочий лист с именем, указанным в [табл. 1.1](#).

3.4. Скопировать из соответствующей таблицы заданий строку с условием задачи ИВ и вставить ее в виде рисунка на лист Excel.

3.5. Составить и набрать в модуле Module1 программу решения задачи ИВ, оформив ее отдельной процедурой с именем, указанным в [табл. 1.1](#). В программе:

а) выбрать рабочий лист, указанный в п. 3.3;

б) очистить содержимое необходимого диапазона ячеек;

в) данные на листе Excel разместить, начиная со строки с номером ИВ + 10;

г) исходные данные и полученные результаты вывести с соответствующими текстовыми пояснениями;

д) числовые результаты вывести в формате контрольных значений, указанных в условии задачи ИВ;

е) при цветовом оформлении вывода данных на лист Excel оттенок заданного базового цвета в модели *RGB* установить равным $200 + \text{ИВ}$;

ж) для каждого оператора предусмотреть комментарии, поясняющие выполняемые действия.

3.6. Запустить программу на исполнение, получить результаты и сверить их с заданными контрольными значениями. При необходимости доработать и отладить программу.

3.7. Записать отлаженную программу в тетрадь.

Таблица 1.1

Требования к именам объектов

Объект	Структура имени	Пояснения	Примеры
Файл (рабочая книга Excel)	Фамилия_NN_Вид работы N.xlsm	<i>Фамилия</i> – фамилия студента; <i>NN</i> – порядковый номер занятия в семестре; <i>Вид работы</i> – лабораторная работа (лаб) или контроль самостоятельной работы (КСР); <i>N</i> – номер занятия по виду работы	Иванов_08_лаб 6.xlsm Иванов_10_КСР 3.xlsm
Рабочий лист Excel	Фамилия_Тема_зN_вN	<i>Тема</i> – краткое обозначение темы задания; <i>зN</i> – номер задания в работе; <i>вN</i> – номер варианта выполняемого задания	Иванов_ОМ_31_в5
Процедура	Фамилия_Тема_зN_вN	Аналогично рабочему листу Excel	Иванов_ОМ_31_в5
	Фамилия_Тема_зN_вN_способN	Выполнение задания <i>способом N</i> (при наличии нескольких способов решения задачи)	Иванов_ОМ_31_в5_способ1

Примечание. Номера заданий (*зN*) должны строго соответствовать их порядковым номерам в перечне заданий на текущую работу.

2. ПОНЯТИЕ МАССИВА ДАННЫХ И ОБЩИЕ ПРИНЦИПЫ ОБРАБОТКИ ЕГО ЭЛЕМЕНТОВ

Во многих вычислительных задачах возникает необходимость манипулировать множеством значений одной переменной с сохранением их в памяти ЭВМ. Порядковый номер очередного значения переменной называется *индексом*. Переменные с индексами являются элементами массивов.

Массив – упорядоченный по возрастанию индексов набор значений одной переменной. Массивы бывают одномерные, двумерные и многомерные. Рассмотрим особенности обработки таких структур на примере одномерных массивов.

В программировании на VBA индекс элемента массива указывается в круглых скобках после имени переменной, он может принимать целые неотрицательные значения и может быть записан в виде константы, переменной или арифметического выражения. Отдельными элементами массива можно пользоваться так же, как и простыми переменными, не забывая при этом указывать индекс, например:

$x(5) = 2$, $r(a) = 2 * w + 4$, $x(p + 7) = y(k) - 1$, `Debug.Print x(n)`.

По умолчанию элементы массива в VBA нумеруются, начиная с нуля. Чтобы установить нумерацию элементов всех массивов с единицы, в программе в разделе общих объявлений модуля необходимо записать оператор `Option Base 1`.

До первого использования массива в программе его необходимо объявить с помощью оператора `Dim` с указанием имени массива, максимально возможного индекса и типа данных в нем:

`Dim Имя_массива(Max_индекс) As Тип`

При этом в качестве максимального индекса могут быть указаны только конкретное числовое значение или объявленная ранее константа, например:

`Dim A(7) As Single` или `Const n As Byte = 7 : Dim A(n) As Single`

Обычно для работы с элементами массива организуют арифметический цикл, параметром которого является индекс элемента массива. В зависимости от специфики задачи в одном цикле можно совмещать ввод, вывод и другие действия с массивом. При необходимости для разных операций с элементами массива могут быть организованы отдельные циклы. Например, если исходные значения элементов массива изменялись в процессе расчета, то для их контроля следует предусмотреть отдельный цикл вывода результирующего массива.

Пример 1. Составить ГСА и программу решения задачи с использованием одномерного массива X по заданию из [табл. 2.1](#). Исходные данные для массива X : $-5,2; 3; -8; -15; -24; -75,4; 81,6; -22; 4,5; -31,7; 43,4; 16$.

Таблица 2.1

Задание для примера 1

Вариант	Условие задачи	Результат
вN	Вычислить количество k , сумму S и среднее арифметическое Sr элементов массива X , значения которых кратны 4. Записать полученное значение Sr на место последнего элемента массива X . Исходный и результирующий массивы вывести в отдельных строках на лист Excel	$k = 5$ $S = -43,20$ $Sr = -8,64$

ГСА решения примера 1 с описанием выполняемых действий приведена на [рис. 2.1](#). Массив X содержит 12 элементов, для работы с ним организован арифметический цикл с параметром i , определяющим индекс элемента массива. В заголовке цикла (блок 3) осуществляется последовательный перебор индексов от первого до последнего (от 1 до 12) с шагом 1. Для контроля правильности заполнения массива сразу после ввода каждого элемента X_i (блок 4) предусмотрен его вывод (блок 5). В блоке 7 реализован вывод элементов, соответствующих проверяемому условию. Поскольку в процессе расчета последний элемент массива изменяется (блок 11), то в ГСА организован еще один цикл для вывода результирующего массива (блоки 12, 13).

В соответствии с ГСА составим программу расчета. Массив X заполним исходными данными, расположенными на листе Excel в 10-й строке (данные обязательно должны быть в ячейках Excel до запуска программы). Выведем на лист Excel в отдельных строках индексы элементов и соответствующие им значения элементов массива; элементы массива, кратные 4; искомые значения k , S , Sr ; результирующий массив.

Для наглядного представления и контроля полученных результатов выделим значение Sr шрифтом красного цвета RGB(200, 0, 0), значение последнего элемента массива – заливкой зеленым цветом RGB(0, 200, 0).

Листинг программы решения примера 1 с подробными комментариями приведен на [рис. 2.2](#), результат ее работы – на [рис. 2.3](#). Сравнение полученных результатов с указанными в задании контрольными значениями позволяет сделать вывод о том, что задача решена верно.




Рис. 2.1. ГСА решения примера 1

Option Explicit	'Запрет использования необъявленных переменных
Option Base 1	'Установка нумерации элементов массива, начиная с единицы

```

Sub Фамилия_ОМ_з1_вN()                                'Начало процедуры

'Объявление переменных
Dim X(12) As Single                                     'Массив X размером 12 элементов
Dim i As Byte                                           'Индекс элемента массива (параметр цикла)
Dim k As Byte, S As Single, Sr As Single               'Количество, сумма, среднее арифметическое
Dim nstr As Byte                                         'Номер строки на листе Excel

Worksheets("Фамилия_ОМ_з1_вN").Select 'Выбор рабочего листа Excel
Range("11:40").Clear                                'Очистка строк с 11 по 40

k = 0 : S = 0                                           'Начальные значения k, S

nstr = 12                                              'Номер начальной строки для вывода результатов
Cells(nstr, 1) = "Работа с массивом X"               'Вывод текстового сообщения

For i = 1 To 12                                        'Заголовок (начало) цикла
    X(i) = Cells(10, i)                               'Ввод значения X(i) из 10-й строки листа Excel
    Cells(nstr + 1, i) = i : Cells(nstr + 2, i) = X(i) 'Вывод индекса i и значения X(i)

    If X(i) Mod 4 = 0 Then                             'Проверка значения X(i) на кратность 4
        Cells(nstr + 4, i) = X(i)                     'Вывод X(i), кратного 4
        k = k + 1 : S = S + X(i)                     'Накопление количества, суммы X(i), кратных 4
    End If
Next                                                    'Возврат к началу цикла

Sr = S / k                                             'Вычисление среднего арифметического значений X(i), кратных 4

'Вывод результирующих значений количества, суммы, среднего арифметического
Cells(nstr + 6, 1) = "k = " : Cells(nstr + 6, 2).NumberFormat = "0" : Cells(nstr + 6, 2) = k
Cells(nstr + 7, 1) = "S = " : Cells(nstr + 7, 2).NumberFormat = "0.00" : Cells(nstr + 7, 2) = S
Cells(nstr + 8, 1) = "Sr = " : Cells(nstr + 8, 2).NumberFormat = "0.00" : Cells(nstr + 8, 2) = Sr
Cells(nstr + 8, 2).Font.Color = RGB(200, 0, 0)       'Выделение значения Sr шрифтом красного цвета

X(12) = Sr                                             'Присвоение последнему элементу массива X значения Sr
Cells(nstr + 10, 1) = "Результирующий массив X"     'Вывод текстового сообщения
For i = 1 To 12                                       'Цикл вывода результирующего массива
    Cells(nstr + 11, i) = X(i)
Next

Cells(nstr + 11, 12).Interior.Color = RGB(0, 200, 0) 'Заливка ячейки с элементом X(12) зеленым цветом

End Sub                                              'Конец процедуры

```

Рис. 2.2. Листинг программы решения примера 1

	A	B	C	D	E	F	G	H	I	J	K	L	M
3	Вариант	Условие задачи										Результат	
4													
5	в.Н	Вычислить количество k , сумму S и среднее арифметическое Sr элементов массива X , значения которых кратны 4. Записать полученное значение Sr на место последнего элемента массива X . Исходный и результирующий массивы вывести в отдельных строках на лист Excel											
6													
7													
8													
9													
10	-5,2	3	-8	-15	-24	-75,4	81,6	-22	4,5	-31,7	43,4	16	Исх. данные
11													
12	Работа с массивом X												
13	1	2	3	4	5	6	7	8	9	10	11	12	Индекс i
14	-5,2	3	-8	-15	-24	-75,4	81,6	-22	4,5	-31,7	43,4	16	Значения X_i
15													
16			-8		-24				4,5	-31,7		16	Знач. X_i , крат. 4
17													
18	$k =$	5											
19	$S =$	-43,20											
20	$Sr =$	-8,64											
21													
22	Результирующий массив X												
23	-5,2	3	-8	-15	-24	-75,4	81,6	-22	4,5	-31,7	43,4	-8,64	
24													

Рис. 2.3. Результат решения примера 1

3. ОПРЕДЕЛЕНИЕ МАКСИМАЛЬНОГО ИЛИ МИНИМАЛЬНОГО ЭЛЕМЕНТА МАССИВА

Одной из наиболее распространенных задач обработки массивов является поиск максимального (минимального) элемента. Процесс ее решения аналогичен определению экстремума функции, подробно рассмотренному в работе [3, п. 8.2.2]. Решение указанной задачи заключается в последовательном сравнении очередного значения элемента массива с текущим значением экстремума, которое при выполнении соответствующего условия становится равным проверяемому элементу массива. При этом обычно в качестве начального максимума принимают значение, заведомо меньше всех элементов массива, в качестве начального минимума – значение, заведомо больше всех элементов массива.

Пример 2. Составить ГСА и программу решения задачи с использованием целочисленного одномерного массива данных по заданию из табл. 3.1. Массив сформировать случайным образом в заданном диапазоне значений.

Задание для примера 2

Вариант	Массив	Диапазон значений элементов массива	Условие задачи
вN	$F(15)$	$-33...22$	Найти максимальный среди неотрицательных элементов массива и его порядковый номер

ГСА решения примера 2 с описанием выполняемых действий приведена на [рис. 3.1](#). В качестве начального максимума примем значение, заведомо меньше всех элементов массива, например, $F_{max} = -34$.

В соответствии с ГСА составим программу расчета. Массив F заполним целыми числами в диапазоне от -33 до 22 с помощью генератора случайных чисел – функции Rnd . В отдельных столбцах Excel выведем порядковые номера и соответствующие им значения всех элементов массива, а также значения неотрицательных элементов. Искомые значения (максимальный среди неотрицательных элементов и его индекс) выведем в строке рядом с соответствующим элементом массива. Для наглядного представления и контроля полученных результатов выделим их заливкой зеленым цветом RGB(0, 200, 0).

Листинг программы решения примера 2 с подробными комментариями приведен на [рис. 3.2](#), результат ее работы – на [рис. 3.3](#). Сравнение полученных результатов с указанными в задании контрольными значениями позволяет сделать вывод о том, что задача решена верно.

Для проверки работоспособности программы вычислим искомое значение максимума с помощью встроенной функции Excel МАКС. Для этого в любой свободной ячейке, например H11, введем формулу (вручную или с помощью мастера функций):

`=МАКС(ЕСЛИ(B12:B26 >= 0; B12:B26; -999))`

Поскольку указанная формула применяется к массиву данных, то после ее ввода нужно обязательно нажать комбинацию клавиш Ctrl + Shift + Enter (в результате формула будет заключена в фигурные скобки). В этой формуле число -999 предусмотрено как сигнальное на случай, если среди элементов массива не окажется значений, соответствующих проверяемому условию. При вычислении минимума в качестве сигнального следует принять число 999.

Значение максимума, вычисленное с помощью функции МАКС, совпадает с программным расчетом, что подтверждает правильность решения задачи.




Рис. 3.1. ГСА решения примера 2


```

Sub Фамилия_ОМ_з2_вN()           'Начало процедуры

'Объявление переменных
Dim F(15) As Integer             'Массив F размером 15 элементов
Dim i As Byte                    'Индекс элемента массива (параметр цикла)
Dim Fmax As Integer              'Максимальный элемент массива
Dim imax As Byte                 'Индекс максимального элемента массива

Worksheets("Фамилия_ОМ_з2_вN").Select 'Выбор рабочего листа Excel
Range("A:G").Clear               'Очистка столбцов с А по G

'Вывод текстового сообщения и заголовков столбцов
Cells(10, 1) = "Работа с массивом"
Cells(11, 1) = "Индекс i": Cells(11, 2) = "F(i)": Cells(11, 3) = "F(i)>=0"

Randomize Timer                  'Установка базы генератора случайных чисел

Fmax = -34: imax = 0             'Начальные значения Fmax и imax

For i = 1 To 15                  'Заголовок (начало) цикла
    F(i) = Fix(Rnd * 56 - 33)     'Заполнение i-го элемента массива F случ. числом от -33 до 22
    'Вывод индекса элемента массива i в 1-й столбец, значения F(i) – во 2-й столбец
    Cells(11 + i, 1) = i: Cells(11 + i, 2) = F(i)

    If F(i) >= 0 Then             'Проверка условия (неотрицательный элемент массива)
        Cells(11 + i, 3) = F(i)  'Вывод неотрицательного значения F(i) в 3-й столбец

        If F(i) > Fmax Then       'Проверка условия (максимальный элемент массива)
            Fmax = F(i)          'Сохранение максимального значения элемента массива и
            imax = i              'его порядкового номера (индекса)
        End If
    End If

Next                             'Возврат к началу цикла

If imax > 0 Then                  'Проверка результата решения задачи
    'Вывод результатов в той же строке, где находится искомый элемент
    Cells(11 + imax, 4) = "Fmax=": Cells(11 + imax, 5) = Fmax
    Cells(11 + imax, 6) = "imax=": Cells(11 + imax, 7) = imax
    'Заливка зеленым цветом диапазона ячеек с результатами расчета
    Range(Cells(11 + imax, 4), Cells(11 + imax, 7)).Interior.Color = RGB(0, 200, 0)
Else
    'Вывод в окно отладки сообщения в случае отсутствия в массиве элементов,
    'удовлетворяющих условию задачи
    MsgBox "imax=" & imax & " В массиве нет элементов по условию задачи"
End If

End Sub                          'Конец процедуры

```

Рис. 3.2. Листинг программы решения примера 2

	A	B	C	D	E	F	G	H	I	J
3										
4	Ва- ри- ант	Мас- сив	Диапазон зна- чений элемен- тов массива	Условие задачи						
7	вN	F(15)	-33...22	Найти максимальный среди неотрицательных элементов массива и его порядковый номер						
9										
10	Работа с массивом							Встр. функция Excel		
11	Индекс i	F(i)	F(i)>=0					21		
12	1	11	11							
13	2	19	19							
14	3	-5								
15	4	-26								
16	5	8	8							
17	6	18	18							
18	7	-24								
19	8	-11								
20	9	-11								
21	10	21	21	Fmax=	21	imax=	10			
22	11	-9								
23	12	0	0							
24	13	-32								
25	14	4	4							
26	15	12	12							
27										

Рис. 3.3. Результат решения примера 2

Примеры формул Excel для обработки массива значений, расположенного в ячейках B1:B8:

для определения минимального среди значений, превышающих число 5:

=МИН(ЕСЛИ(B1:B8 > 5; B1:B8; 999))

для определения максимального среди значений, кратных числу 7:

=МАКС(ЕСЛИ(ОСТАТ(B1:B8; 7) = 0; B1:B8; -999))

для определения максимального среди значений с нечетными индексами (индексы находятся в ячейках A1:A8):

=МАКС(ЕСЛИ(ОСТАТ(A1:A8; 2) = 1; B1:B8; -999))

для определения минимального среди значений, превышающих свой индекс не менее чем на 22 (индексы находятся в ячейках A1:A8):

=МИН(ЕСЛИ(ABS(B1:B8 - A1:A8) >= 22; B1:B8; 999))

4. ЗАДАНИЯ

Задание 1. В соответствии с общими требованиями к выполнению заданий (см. [разд. 1](#)) и индивидуальным вариантом ([табл. 4.1](#)) составить ГСА и программу решения задачи с использованием одномерного массива F . Полученный результат записать на место элемента массива с индексом, равным номеру варианта (для варианта 0 – на место первого элемента массива).

Исходные данные для массива F : $-5,2; 3; -8; -15; -24; -75,4; 81,6; -22; 4,5; -31,7; 43,4; 16; -5; 11,3; 24,7; -42; 15; 1,4$.

Работу выполнить и оформить по образцу [примера 1](#). Краткое обозначение темы задания 1 в именах объектов – ОМ.

Таблица 4.1

Индивидуальные варианты задания 1

Вариант	Условие задачи	Результат
1	2	3
0	Определить количество элементов массива F , значения которых больше 2	8
1	Вычислить произведение элементов массива F , целая часть которых кратна 8	75878,41
2	Вычислить среднее арифметическое положительных элементов массива F	22,32
3	Определить количество элементов массива F с индексами, кратными 4	4
4	Вычислить сумму элементов массива F , значения которых меньше 3	-226,9
5	Определить количество элементов массива F , имеющих целые значения	9
6	Вычислить сумму элементов массива F с индексами, кратными 3	-36,8
7	Вычислить сумму элементов массива F , дробная часть которых меньше 0,5	-106,5
8	Вычислить среднее арифметическое элементов массива F с четными индексами	-17,16
9	Определить количество элементов массива F , дробная часть которых больше 0,3	8
10	Вычислить сумму элементов массива F , значения которых не превышают своего индекса	-196,1

Окончание табл. 4.1

1	2	3
11	Вычислить произведение элементов массива F , значения которых больше 20	87473,6
12	Вычислить среднее арифметическое элементов массива F , значения которых меньше 17	-11,81
13	Вычислить сумму элементов массива F , имеющих дробные значения	54,6
14	Определить количество элементов массива F , целая часть которых кратна 3	8
15	Вычислить сумму элементов массива F , значения которых не превышают -30	-149,1
16	Вычислить сумму квадратов отрицательных элементов массива F	9855,09
17	Вычислить среднее арифметическое элементов массива F , имеющих целые значения	-9,11
18	Вычислить сумму целых частей положительных элементов массива F	198

Задание 2. В соответствии с общими требованиями к выполнению заданий (см. [разд. 1](#)) и индивидуальным вариантом ([табл. 4.2](#)) составить ГСА и программу решения задачи с использованием целочисленного одномерного массива данных. Работу выполнить и оформить по образцу [примера 2](#). Краткое обозначение темы задания 2 в именах объектов – ОМ.

Таблица 4.2

Индивидуальные варианты задания 2

Вариант	Массив	Диапазон значений элементов массива	Условие задачи
1	2	3	4
0	$W(11)$	-35...25	Найти минимальный среди элементов массива, значения которых больше 2, и его порядковый номер
1	$Y(25)$	-22...44	Найти максимальный среди элементов массива, значения которых кратны 8, и его порядковый номер
2	$D(15)$	-20...25	Найти минимальный среди элементов массива с четными индексами и его порядковый номер
3	$K(17)$	-40...40	Найти максимальный среди элементов массива, значения которых меньше 17, и его порядковый номер

1	2	3	4
4	$F(22)$	$-70 \dots -30$	Найти минимальный среди элементов массива с индексами, кратными 3, и его порядковый номер
5	$T(18)$	$10 \dots 75$	Найти максимальный среди элементов массива, значения которых не превышают 50, и его порядковый номер
6	$H(16)$	$-25 \dots 10$	Найти максимальный среди отрицательных элементов массива и его порядковый номер
7	$N(26)$	$65 \dots 150$	Найти минимальный среди элементов массива, значения которых кратны 4, и его порядковый номер
8	$A(10)$	$-20 \dots 35$	Найти максимальный среди элементов массива, значения которых не превышают свой индекс, и его порядковый номер
9	$P(17)$	$-53 \dots 15$	Найти минимальный среди нечетных элементов массива и его порядковый номер
10	$R(14)$	$-15 \dots 45$	Найти максимальный среди элементов массива, значения которых не превышают 20, и его порядковый номер
11	$C(25)$	$14 \dots 86$	Найти минимальный среди элементов массива с индексами, кратными 4, и его порядковый номер
12	$L(17)$	$-33 \dots 33$	Найти максимальный среди четных элементов массива и его порядковый номер
13	$Z(18)$	$-30 \dots 60$	Найти минимальный среди положительных элементов массива и его порядковый номер
14	$B(23)$	$-25 \dots 15$	Найти максимальный среди элементов массива, значения которых не превышают -15 , и его порядковый номер
15	$E(14)$	$-11 \dots 22$	Найти максимальный среди элементов массива с нечетными индексами и его порядковый номер
16	$V(24)$	$-10 \dots 55$	Найти минимальный среди элементов массива, значения которых превышают свой индекс более чем в два раза, и его порядковый номер
17	$U(19)$	$-35 \dots 25$	Найти минимальный среди элементов массива, значения которых кратны 3, и его порядковый номер
18	$M(20)$	$85 \dots 105$	Найти минимальный среди элементов массива, значения которых кратны своему индексу, и его порядковый номер

5. КОНТРОЛЬНЫЕ ВОПРОСЫ

- 1) Что такое массив данных?
- 2) Какие значения могут принимать индексы элементов массива?
- 3) Какая базовая алгоритмическая структура обязательно используется при обработке элементов массива?
- 4) Перечислите способы ввода и вывода элементов массива.
- 5) Можно ли в одном цикле совмещать ввод и вывод элементов массива?
- 6) Как сформировать массив случайных чисел?

6. ПРИМЕРЫ ТЕСТОВЫХ ВОПРОСОВ

Вопрос № 1 (один верный ответ)

Массив – это ...

Варианты ответов:

- 1) упорядоченный по возрастанию индексов набор значений одной переменной;
- 2) ограниченная двоеточием последовательность любых символов;
- 3) совокупность разнородных данных, описываемых и обрабатываемых как единое целое;
- 4) именованный набор однотипных данных на диске.

Вопрос № 2 (один верный ответ)

Как осуществляется обращение к отдельному элементу массива?

Варианты ответов:

- 1) по имени массива и номеру элемента;
- 2) по имени массива и значению элемента;
- 3) по номеру элемента с указанием типа данных;
- 4) обращение к отдельному элементу массива невозможно.

Вопрос № 3 (несколько верных ответов)

Массив характеризуется ...

Варианты ответов:

- 1) именем;
- 2) количеством измерений;
- 3) размером;
- 4) количеством обращений.

Вопрос № 4 (один верный ответ)

Как записать элемент массива X с порядковым номером 5?

Варианты ответов:

- 1) $X(5)$;
- 2) $X5$;
- 3) $X№5$;
- 4) X_5 .

Вопрос № 5 (один верный ответ)

Что произойдет в результате выполнения строк программного кода?

$w = X(z)$

$X(z) = X(z + 1)$

$X(z + 1) = w$

Варианты ответов:

- 1) обмен значений элементов массива $X(z)$ и $X(z + 1)$;
- 2) присвоение элементам массива $X(z)$ и $X(z + 1)$ значения переменной w ;
- 3) удаление элемента массива $X(z)$.

Вопрос № 6 (один верный ответ)

Что произойдет в результате выполнения строк программного кода?

```
Dim k As Integer  
Dim X(6) As Integer  
For k = 1 To 6  
    X(k) = Cells(3, k)  
Next k
```

Варианты ответов:

- 1) ввод значений массива X из 3-й строки листа Excel;
- 2) вывод значений массива X в 3-ю строку листа Excel;
- 3) программа записана с ошибкой и не может быть выполнена.

Библиографический список

1. Лебедев, В. М. Программирование на VBA в MS Excel : учебное пособие / В. М. Лебедев. – Москва : Юрайт, 2020. – 306 с. – Текст : непосредственный.
2. Казанский, А. А. Прикладное программирование на Excel 2019 : учебное пособие / А. А. Казанский. – Москва : Юрайт, 2020. – 171 с. – Текст : непосредственный.
3. Сидорова, Е. А. Основы программирования на языке VBA : учебное пособие / Е. А. Сидорова, С. П. Железняк. – Омск : Омский государственный университет путей сообщения, 2021. – 118 с. – Текст : непосредственный.
4. Сидорова, Е. А. Программирование арифметических циклов на VBA : учебно-методическое пособие / Е. А. Сидорова, А. В. Долгова, С. П. Железняк. – Омск : Омский государственный университет путей сообщения, 2021. – 30 с. – Текст : непосредственный.
5. ГОСТ 19.701–90 (ИСО 5807–85). Единая система программной документации. Схемы алгоритмов, программ, данных и систем. Обозначения условные и правила выполнения. – Москва : Изд-во стандартов, 1990. – 36 с. – Текст : непосредственный.

Учебное издание

СИДОРОВА Елена Анатольевна,
ЖЕЛЕЗНЯК Светлана Петровна

ОБРАБОТКА ОДНОМЕРНЫХ МАССИВОВ НА VBA

Учебно-методическое пособие

Редактор Н. А. Майорова

Подписано в печать 17.01.2022. Формат $60 \times 84 \frac{1}{16}$.
Офсетная печать. Бумага офсетная. Усл. печ. л. 1,4. Уч.-изд. л. 1,6.
Тираж 100 экз. Заказ .

**

Редакционно-издательский отдел ОмГУПСа
Типография ОмГУПСа

*

644046, г. Омск, пр. Маркса, 35